
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 22.1.1.M3 which is part of the NBIOT test suite in the ‘iwd-TTCN3-B2017-06_D18wk12’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
22.1.1.M3
Test Group:
\NBIOT\22_1\NBIOT_General_CP.ttcn
ATS Version:
iwd-TTCN3-B2017-06_D18wk12
System Simulator used:
Anritsu Protocol Conformance Test System ME7834LA




UE used:
Qualcomm MDM9206
Verification Status:
PASS


4. Corrections required for NB-IoT TC 22.1.1.M3
Change 1

	Function name
	f_TC_22_1_1_Module1

	Reason for change
	1. Variable v_AttachWithoutPDN is populated the opposite way based on the px_DoAttachWithoutPDN.
2. Wrong cell is used in step 5 and 6.

3. APN element will not get populated in case UE does not trigger ESM information exchange. Not including APN during default bearer establishment will lead to reject of the default bearer by UE.

	Summary of change
	1. Corrected way of populating v_AttachWithoutPDN.
2. Replaced Ncell1 with Ncell50 in steps 5 and 6 of Module 1.
3. Default APN from px_AccessPointName is populated in step 11a2 in case UE did not send ESM information.

	TTCN module
	NBIOT_General_CP.ttcn

	MCC160 Comment
	1,2. Accepted
3. Rejected.  In this case it will be populated from the PDN Connectivity Req message


Before  change

    function f_TC_22_1_1_Module1() runs on NBIOT_PTC

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_UL_Message_Type v_AttachReq;

    var template (present) NAS_UL_Message_Type v_ESMMessage;

    var template (present) B2_Type v_PNB := '01'B;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var AccessPointName v_APN;

    var EUTRA_ASN1_TrackingAreaCode_Type v_TAC := f_NBIOT_CellInfo_GetTAC(nbiot_Cell50);

    var template (value) PDN_Address v_PDN_Address;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(nbiot_Cell50);

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var NasCount_Type v_NasCountUL;

    var NBIOT_SecurityParams_Type v_AuthParams;

    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(nbiot_Cell50);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;
    var template (present) NAS_UL_Message_Type v_ESMMessage_BearerAccept;

    //@siclog "Step 1" siclog@

    f_UT_SwitchOnUE(UT, false);

    //@siclog "Step 2" siclog@

    f_NBIOT_RRC_ConnectionRequest_Def(nbiot_Cell50, mo_Signalling);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step2");

    //@siclog "Step 3" siclog@

    f_NBIOT_RRC_ConnectionSetup_Def();

    //@siclog "Step 4" siclog@

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;
    }

    if (px_DoAttachWithoutPDN) {

      v_ESMMessage := cr_ESM_DUMMY_MESSAGE;

      _;   

    } else {
      v_AttachWithoutPDN := true_;
      v_ESMMessage := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                       v_PDNType,

                                                       -,

                                                       v_HeaderCompressConfigRX);

    }

    if (pc_User_Plane_CIoT_Optimisation) {

      v_PNB := ('01'B, '10'B);

    }

    v_AttachReq := cr_CIOT_ATTACH_REQUEST(cr_AdditionalUpdateType_CIOT(v_PNB, f_ConvertBoolToBit (pc_NB_S1_only and px_SMSTransport_CP_CIoT)));

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(nbiot_Cell50,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                             v_AttachReq,

                                                                                             v_ESMMessage));

    if (not f_Check_UE_NetworkCap_CIOT (v_NasInd_Rxd.Pdu.Msg.aTTACH_REQUEST.ueNetworkCapability, CIOT_CP)) {

      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "UE Network Cap failed CIOT bit check");

    }

    if (not px_DoAttachWithoutPDN) {

      v_PDNConnectivityReq_Rxd := v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg;

      f_NBIOT_MobileInfo_SetAssignedPdnAndPTI (0, // This is registration, so this is the first PDN

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue,

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier);

      if (ispresent(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName)) {

        v_APN := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName;

        v_APN.iei := omit;

        f_NBIOT_MobileInfo_SetAPN (0, v_APN);

      }

      f_NBIOT_MobileInfo_SetExtPCO (0, v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions);

    }

    // If something wasn't correct in this message, it would have failed before now

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step4a1/4b1");

    //+   The SS Transmits an IDENTITY REQUEST message

    //@siclog "Step 5" siclog@

    SRB.send(cas_NB_SRB1bis_NasPdu_REQ(nbiot_Cell1,

                                       cs_TimingInfo_Now,

                                       cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_IDENTITY_REQUEST(tsc_IdType_IMSI))));

    //+   Check: Does the UE transmit an IDENTIY RESPONSE message

    //@siclog "Step 6" siclog@

    SRB.receive(car_NB_SRB1bis_NasPdu_IND(nbiot_Cell1,

                                          cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_IDENTITY_RESPONSE(f_Imsi2MobileIdentity(px_IMSI_Def)))));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    // Set UE network capabilities as sent by the UE to be used later (see TS 24.301 clause 5.4.3.3)

    v_AuthParams := f_NBIOT_Security_Get();

    v_AuthParams.NAS_SecurityCap := f_EUTRA_NB_FillNAS_SecurityCapability(v_NasInd_Rxd.Pdu.Msg.aTTACH_REQUEST);

    f_NBIOT_Security_Set(v_AuthParams);

    //@siclog "Step 7-8" siclog@

    v_AuthParams := f_NBIOT_NAS_Authentication(nbiot_Cell50,

                                               tsc_SRB1bis,

                                               v_AuthParams,

                                               tsc_SHT_IntegrityProtected_Ciphered,

                                               tsc_SHT_IntegrityProtected_Ciphered,

                                               v_PLMN);                        // AUTHENTICATION REQUEST & RESPONSE

    f_NBIOT_Security_Set(v_AuthParams);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    //@siclog "Step 9-10" siclog@

    v_NasCountUL := f_NBIOT_NAS_ActivateSecurity(nbiot_Cell50, v_AuthParams);   // NAS SECURITY COMMAND & COMPLETE

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 10");

    if (not px_DoAttachWithoutPDN) { // PDN Requested

      if (f_CheckEsmInfoTransferFlag(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST)) {

        //@siclog "Step 11a1" siclog@

        SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                       tsc_SRB1bis,

                                       cs_TimingInfo_Now,

                                       cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_Esm_Information_Request(f_NBIOT_MobileInfo_GetPTI(0)))));

        //@siclog "Step 11a2" siclog@

        SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                          tsc_SRB1bis,

                                          cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cdr_CIOT_Esm_Information_Response(f_NBIOT_MobileInfo_GetPTI(0), cr_ExtdPCOAny))))

          -> value v_ReceivedAsp;

        if (ispresent(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.accessPointName)) {

          v_APN := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.accessPointName;

          v_APN.iei := omit;

          f_NBIOT_MobileInfo_SetAPN (0, v_APN);

        }

        f_NBIOT_MobileInfo_SetExtPCO (0, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.extdProtocolConfigurationOptions); // Table 22.1.1.3-31 specifies mandatory

      }

      v_PDN_Address := f_CIOT_GetPDNAddress(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, f_CheckExtdPCOforIPallocationViaNas(f_NBIOT_MobileInfo_GetExtdPCO(0)));

      v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(f_NBIOT_MobileInfo_GetExtdPCO(0)));

      //@siclog "Step 12b1" siclog@

      SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                     tsc_SRB1bis,

                                     cs_TimingInfo_Now,

                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                    cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),

                                                                                          v_Guti,

                                                                                          cs_EPS_NwkFtSupport_CP_LCS_XtdPCO,

                                                                                          cs_AdditionalUpdateResult_CP_SMS,

                                                                                          -,

                                                                                          omit,

                                                                                          omit,

                                                                                          omit),

                                                                    cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                                              f_NBIOT_MobileInfo_GetPTI(0),

                                                                                                              v_APN,

                                                                                                              v_PDN_Address,

                                                                                                              cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),

                                                                                                              -,

                                                                                                              -,

                                                                                                              -,

                                                                                                              v_HeaderCompressConfigTX,
                                                                                                              cs_ControlPlaneInd,

                                                                                                              cs_ServingPLMNRateControl(int2oct (10, 2))))));

      v_ESMMessage_BearerAccept := cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId);

    } else { // No PDN

      //@siclog "Step 12a1" siclog@

      SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                     tsc_SRB1bis,

                                     cs_TimingInfo_Now,

                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                    cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),

                                                                                          v_Guti,

                                                                                          cs_EPS_NwkFtSupport_CP_LCS_XtdPCO,

                                                                                          cs_AdditionalUpdateResult_CP_SMS,

                                                                                          -,

                                                                                          omit,

                                                                                          omit,

                                                                                          omit),

                                                                    cs_ESM_DUMMY_MESSAGE(v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.eSM_DUMMY_MESSAGE.procedureTransactionIdentifier))));

      v_ESMMessage_BearerAccept := cr_ESM_DUMMY_MESSAGE;

    }

    //@siclog "Step 13a1/13b1" siclog@

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                      tsc_SRB1bis,

                                      cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                        cr_508_ATTACH_COMPLETE,

                                                                        v_ESMMessage_BearerAccept)));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 13");

    if (not px_DoAttachWithoutPDN) { // PDN Requested

      if (not match (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, '101'B)) { // So IP PDN

        f_NBIOT_DelayForUserPlaneSignalling_CP(nbiot_Cell50);

      }

    }

    //@siclog "Step 14" siclog@

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell50);

  } // End of Module 1
After  change

    function f_TC_22_1_1_Module1() runs on NBIOT_PTC

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_UL_Message_Type v_AttachReq;

    var template (present) NAS_UL_Message_Type v_ESMMessage;

    var template (present) B2_Type v_PNB := '01'B;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var AccessPointName v_APN;

    var EUTRA_ASN1_TrackingAreaCode_Type v_TAC := f_NBIOT_CellInfo_GetTAC(nbiot_Cell50);

    var template (value) PDN_Address v_PDN_Address;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(nbiot_Cell50);

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var NasCount_Type v_NasCountUL;

    var NBIOT_SecurityParams_Type v_AuthParams;

    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(nbiot_Cell50);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;
    var template (present) NAS_UL_Message_Type v_ESMMessage_BearerAccept;

    //@siclog "Step 1" siclog@

    f_UT_SwitchOnUE(UT, false);

    //@siclog "Step 2" siclog@

    f_NBIOT_RRC_ConnectionRequest_Def(nbiot_Cell50, mo_Signalling);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step2");

    //@siclog "Step 3" siclog@

    f_NBIOT_RRC_ConnectionSetup_Def();

    //@siclog "Step 4" siclog@

    //No header compression will be used as per 36.508 (change from R5-171137)

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;

    }
    if (px_DoAttachWithoutPDN) {

      v_ESMMessage := cr_ESM_DUMMY_MESSAGE;

      v_AttachWithoutPDN := true_;   
    } else {

      v_ESMMessage := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                       v_PDNType,

                                                       -,

                                                       v_HeaderCompressConfigRX);

    }

    if (pc_User_Plane_CIoT_Optimisation) {

      v_PNB := ('01'B, '10'B);

    }

    v_AttachReq := cr_CIOT_ATTACH_REQUEST(cr_AdditionalUpdateType_CIOT(v_PNB, f_ConvertBoolToBit (pc_NB_S1_only and px_SMSTransport_CP_CIoT)));

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(nbiot_Cell50,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                             v_AttachReq,

                                                                                             v_ESMMessage));

    if (not f_Check_UE_NetworkCap_CIOT (v_NasInd_Rxd.Pdu.Msg.aTTACH_REQUEST.ueNetworkCapability, CIOT_CP)) {

      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "UE Network Cap failed CIOT bit check");

    }

    if (not px_DoAttachWithoutPDN) {

      v_PDNConnectivityReq_Rxd := v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg;

      f_NBIOT_MobileInfo_SetAssignedPdnAndPTI (0, // This is registration, so this is the first PDN

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue,

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier);

      if (ispresent(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName)) {

        v_APN := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName;

        v_APN.iei := omit;

        f_NBIOT_MobileInfo_SetAPN (0, v_APN);

      }

      f_NBIOT_MobileInfo_SetExtPCO (0, v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions);

    }

    // If something wasn't correct in this message, it would have failed before now

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step4a1/4b1");

    //+   The SS Transmits an IDENTITY REQUEST message

    //@siclog "Step 5" siclog@

    SRB.send(cas_NB_SRB1bis_NasPdu_REQ(nbiot_Cell50, 

                                       cs_TimingInfo_Now,

                                       cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_IDENTITY_REQUEST(tsc_IdType_IMSI))));

    //+   Check: Does the UE transmit an IDENTIY RESPONSE message

    //@siclog "Step 6" siclog@

    SRB.receive(car_NB_SRB1bis_NasPdu_IND(nbiot_Cell50, 

                                          cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_IDENTITY_RESPONSE(f_Imsi2MobileIdentity(px_IMSI_Def)))));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    // Set UE network capabilities as sent by the UE to be used later (see TS 24.301 clause 5.4.3.3)

    v_AuthParams := f_NBIOT_Security_Get();

    v_AuthParams.NAS_SecurityCap := f_EUTRA_NB_FillNAS_SecurityCapability(v_NasInd_Rxd.Pdu.Msg.aTTACH_REQUEST);

    f_NBIOT_Security_Set(v_AuthParams);

    //@siclog "Step 7-8" siclog@

    v_AuthParams := f_NBIOT_NAS_Authentication(nbiot_Cell50,

                                               tsc_SRB1bis,

                                               v_AuthParams,

                                               tsc_SHT_IntegrityProtected_Ciphered,

                                               tsc_SHT_IntegrityProtected_Ciphered,

                                               v_PLMN);                        // AUTHENTICATION REQUEST & RESPONSE

    f_NBIOT_Security_Set(v_AuthParams);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    //@siclog "Step 9-10" siclog@

    v_NasCountUL := f_NBIOT_NAS_ActivateSecurity(nbiot_Cell50, v_AuthParams);   // NAS SECURITY COMMAND & COMPLETE

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 10");

    if (not px_DoAttachWithoutPDN) { // PDN Requested

      if (f_CheckEsmInfoTransferFlag(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST)) {

        //@siclog "Step 11a1" siclog@

        SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                       tsc_SRB1bis,

                                       cs_TimingInfo_Now,

                                       cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_Esm_Information_Request(f_NBIOT_MobileInfo_GetPTI(0)))));

        //@siclog "Step 11a2" siclog@

        SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                          tsc_SRB1bis,

                                          cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cdr_CIOT_Esm_Information_Response(f_NBIOT_MobileInfo_GetPTI(0), cr_ExtdPCOAny))))

          -> value v_ReceivedAsp;

        if (ispresent(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.accessPointName)) {

          v_APN := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.accessPointName;

          v_APN.iei := omit;

          f_NBIOT_MobileInfo_SetAPN (0, v_APN);

        }

        f_NBIOT_MobileInfo_SetExtPCO (0, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.extdProtocolConfigurationOptions); // Table 22.1.1.3-31 specifies mandatory

      } else {

        v_APN := valueof(cs_AccessPointName(px_AccessPointName));

      }
      v_PDN_Address := f_CIOT_GetPDNAddress(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, f_CheckExtdPCOforIPallocationViaNas(f_NBIOT_MobileInfo_GetExtdPCO(0)));

      v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(f_NBIOT_MobileInfo_GetExtdPCO(0)));

      //@siclog "Step 12b1" siclog@

      SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                     tsc_SRB1bis,

                                     cs_TimingInfo_Now,

                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                    cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),

                                                                                          v_Guti,

                                                                                          cs_EPS_NwkFtSupport_CP_LCS_XtdPCO,

                                                                                          cs_AdditionalUpdateResult_CP_SMS,

                                                                                          -,

                                                                                          omit,

                                                                                          omit,

                                                                                          omit),

                                                                    cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                                              f_NBIOT_MobileInfo_GetPTI(0),

                                                                                                              v_APN,

                                                                                                              v_PDN_Address,

                                                                                                              cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),

                                                                                                              -,

                                                                                                              -,

                                                                                                              -,

                                                                                                              v_HeaderCompressConfigTX, 

                                                                                                              cs_ControlPlaneInd,

                                                                                                              cs_ServingPLMNRateControl(int2oct (10, 2))))));

      v_ESMMessage_BearerAccept := cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId);

    } else { // No PDN

      //@siclog "Step 12a1" siclog@

      SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                     tsc_SRB1bis,

                                     cs_TimingInfo_Now,

                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                    cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),

                                                                                          v_Guti,

                                                                                          cs_EPS_NwkFtSupport_CP_LCS_XtdPCO,

                                                                                          cs_AdditionalUpdateResult_CP_SMS,

                                                                                          -,

                                                                                          omit,

                                                                                          omit,

                                                                                          omit),

                                                                    cs_ESM_DUMMY_MESSAGE(v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.eSM_DUMMY_MESSAGE.procedureTransactionIdentifier))));

      v_ESMMessage_BearerAccept := cr_ESM_DUMMY_MESSAGE;

    }

    //@siclog "Step 13a1/13b1" siclog@

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                      tsc_SRB1bis,

                                      cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                        cr_508_ATTACH_COMPLETE,

                                                                        v_ESMMessage_BearerAccept)));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 13");

    if (not px_DoAttachWithoutPDN) { // PDN Requested

      if (not match (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, '101'B)) { // So IP PDN

        f_NBIOT_DelayForUserPlaneSignalling_CP(nbiot_Cell50);

      }

    }

    //@siclog "Step 14" siclog@

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell50);

  } // End of Module 1
Change 2

	Function name
	f_Check_UE_NetworkCap_CIOT

	Reason for change
	The px_DoAttachWithoutPDN does not reflect capability of the UE, but merely the way the UE is configured. 

Based on 24.301 cl. 5.5.1

If EMM-REGISTERED without PDN connection is supported by the UE and the MME, a default bearer need not be requested by the UE during the attach procedure.

…

If the UE supports CIoT EPS optimizations, it shall indicate in the UE network capability IE of the ATTACH REQUEST message whether it supports EMM-REGISTERED without PDN connection.
UE shall advertise support of EMM-REGISTERED without PDN connection regardless whether it requests PDN connectivity or not. The only criterion is the actual UE support of this CIoT EPS optimisation.  

Prose CR associated with this change (to change 36.523-1 Table 22.1.1.3-4).

	Summary of change
	CIoT optimization capability regarding “attach w/o PDN” is not checked against px_DoAttachWithoutPDN anymore.

	TTCN module
	NBIOT_General_CP.ttcn

	MCC160 Comment
	Accepted


Before  change

    function f_Check_UE_NetworkCap_CIOT(UE_NetworkCap p_UE_NetworkCap,

                                      UE_NetworkCap_ToBeTested p_ToBeTested) return boolean

  {

    var boolean v_Result:= false;

    var template (present) B8_Type v_Expected := ?;

    // var template bitstring v_Test := '010'B & ? & '1'B & ? length (1) & '1'B;

    var template (present) B1_Type v_CP := ?;

    var template (present) B1_Type v_UP := ?;

    select (p_ToBeTested) {

      case (CIOT_CP) {

        v_CP := '1'B;

      }

      case (CIOT_UP) {

        v_UP := '1'B;

      }

    }

    //v_Expected := '1'B & f_ConvertBoolToBit (pc_HCCPCIoT) & f_ConvertBoolToBit(px_DoAttachWithoutPDN) & ? length (1) & v_UP & v_CP & ? length (2);

    v_Expected := replace (v_Expected, 1, 1, f_ConvertBoolToBit (pc_HCCPCIoT));

    v_Expected := replace (v_Expected, 2, 1, f_ConvertBoolToBit (px_DoAttachWithoutPDN));
    v_Expected := replace (v_Expected, 4, 1, v_UP);

    v_Expected := replace (v_Expected, 5, 1, v_CP);

    if (match (oct2bit(p_UE_NetworkCap.cIoT_proSeCap), v_Expected)) {

      v_Result := true;

    }

    return v_Result;

  }
After  change

      function f_Check_UE_NetworkCap_CIOT(UE_NetworkCap p_UE_NetworkCap,

                                      UE_NetworkCap_ToBeTested p_ToBeTested) return boolean

  {

    var boolean v_Result:= false;

    var template (present) B8_Type v_Expected := ?;

    // var template bitstring v_Test := '010'B & ? & '1'B & ? length (1) & '1'B;

    var template (present) B1_Type v_CP := ?;

    var template (present) B1_Type v_UP := ?;

    select (p_ToBeTested) {

      case (CIOT_CP) {

        v_CP := '1'B;

      }

      case (CIOT_UP) {

        v_UP := '1'B;

      }

    }

    //v_Expected := '1'B & f_ConvertBoolToBit (pc_HCCPCIoT) & f_ConvertBoolToBit(px_DoAttachWithoutPDN) & ? length (1) & v_UP & v_CP & ? length (2);

    v_Expected := replace (v_Expected, 1, 1, f_ConvertBoolToBit (pc_HCCPCIoT));

    v_Expected := replace (v_Expected, 4, 1, v_UP);

    v_Expected := replace (v_Expected, 5, 1, v_CP);

    if (match (oct2bit(p_UE_NetworkCap.cIoT_proSeCap), v_Expected)) {

      v_Result := true;

    }

    return v_Result;

  }
Change 3

	Function name
	f_TC_22_1_1_Module3

	Reason for change
	1. The check required in step 16a7 whether the DL SMS and UL SMS match can be applied to TP_Data and not to entire CP PDU. This is based on 36.509 cl. 5.4.4f.1:
if GH_RLC_SDU_loopback is TRUE submit the TPDU received in the RP User data element for each received RP-DATA message in a RLC SDU repeated GH_NUMBER_UL_DATA_REPETITIONS of times to the RLC AM-SAP of SRB1bis for NB-IoT UE or to the RLC AM-SAP of SRB2 for E-UTRA UE for transmission in uplink
For this reason the matching mechanism needs to be changed so it is applied only to TP PDU. For the sake of simplification it is also proposed to move this check outside the function f_NBIOT_508Check_CP_MOSMSDataTransfer.
2. Above change implies that function f_NBIOT_508Check_CP_MOSMSDataTransfer  will not need the “expected” SMS as one of the input parameters.

3. Labels for steps 16a7B,C and D do not correspond to steps in 36.523-1 – this is only formal correction in the commented text.
4. TI values used for CP ACK in step 16a8A need to be taken from the corresponding SMS received in step 16a8 and not from 16a7C.
5. The matching check required by step16a8 is not implemented in current TTCN, therefore extra code to perform check on matching TP PDU should be added.

	Summary of change
	1. Changed the matching for DL and UL SMS used in step 16a7 so TP PDU is checked instead of CP PDU.This check is now performed right after f_NBIOT_508Check_CP_MOSMSDataTransfer function.
2. Function f_NBIOT_508Check_CP_MOSMSDataTransfer is called only with cellId parameter.
3. Correction of commented labels of individual steps 16a7B,C and D.
4. TI values taken from SMS received in step 16a8
5. Added check whether SMS and its TP data received in step 16a8 match with the DL SMS sent in step 16a4.

	TTCN module
	NBIOT_General_CP.ttcn

	MCC160 Comment
	1. Accepted. MCC160 Implementation see below.
2. Rejected. MCC160 Implementation see below.
3. Accepted

4. Accepted. 
5. Accepted


Before  change

  function f_TC_22_1_1_Module3() runs on NBIOT_PTC

  {

    var  octetstring v_SMS_Message;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var octetstring v_Rxd_SMS;

    var NAS_MessageContainer v_NAS_RcvMsgContainer_SMS;

    var template (value) NAS_MessageContainer v_NAS_TxMessageContainer_SMS;

    var CP_PDU_Type v_RcvSMS_CP_PDU;

    var template CP_PDU_Type v_ExpSMS_CP_PDU;

    var bitstring v_NAS_RcvBitsMsgContainer_SMS;

    var B3_Type v_TI_Value;

    var TransactionIdentifier_Type v_MO_TI_FromSS;

    var TransactionIdentifier_Type v_MO_TI_FromUE;

    var O1_Type v_RpMsgRefRcvd;

    var SMS_DELIVER_Type v_SMS_DELIVER_Sent;

    var octetstring v_PRBS_SMS := f_GetN_OctetsFromPRBS (0, 123); // 123 fits with the requirement from 36.509

    var integer v_LoopDelay := 60;

    timer t_LoopDelay;
...

    //@siclog "Step 16a6" siclog@

    t_LoopDelay.timeout;

    //@siclog "Step 16a7" siclog@

    // 1st repetition

    v_Rxd_SMS := f_NBIOT_508Check_CP_MOSMSDataTransfer (nbiot_Cell50, v_SMS_Message);

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_Rxd_SMS);

    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (cr_MT_TI0_FromUE);

    v_RcvSMS_CP_PDU := f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "22.1.1", "Step16a7");
    //@siclog "Step 16a7B" siclog@

    // The SS transmits a CP-ACK encapsulated in a Downlink NAS Transport message.

    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;

    v_MO_TI_FromSS.tI_Flag := '1'B;

    v_MO_TI_FromSS.tI_Value := v_TI_Value;

    v_MO_TI_FromUE.tI_Flag := '0'B;

    v_MO_TI_FromUE.tI_Value := v_TI_Value;

    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;

    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (v_MO_TI_FromSS)));

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          v_NAS_TxMessageContainer_SMS))));

    //@siclog "Step 16a7C" siclog@

    // The SS transmits a CP-DATA containing an RP-ACK RPDU encapsulated in an Downlink NAS transport message

    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (v_MO_TI_FromSS,

                                                                                                cs_CP_UserData_RP_ACK_dl(v_RpMsgRefRcvd))));

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          v_NAS_TxMessageContainer_SMS))));

    //@siclog "Step 16a7D" siclog@

    // Check: Does the UE transmit a CP-ACK encapsulated in an Uplink NAS Transport message?

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                   tsc_SRB1bis,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;

    v_NAS_RcvMsgContainer_SMS :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);

    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (v_MO_TI_FromUE);

    v_RcvSMS_CP_PDU := f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test case 22.1.1", "Step 9C");

    //@siclog "Step 16a8" siclog@

    // 2nd repetition

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                           tsc_SRB1bis,

                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                             cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                               cr_NAS_MsgContainer_SMS (?)))));

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step16a8");

    // Note TI value used by the UE and set TI variables accordingly

    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;

    v_MO_TI_FromSS.tI_Flag := '1'B;

    v_MO_TI_FromSS.tI_Value := v_TI_Value;

    v_MO_TI_FromUE.tI_Flag := '0'B;

    v_MO_TI_FromUE.tI_Value := v_TI_Value;

    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;

    //@siclog "Step 16a8A" siclog@
After  change

  function f_TC_22_1_1_Module3() runs on NBIOT_PTC

  {

    var  octetstring v_SMS_Message;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var octetstring v_Rxd_SMS;

    var NAS_MessageContainer v_NAS_RcvMsgContainer_SMS;

    var template (value) NAS_MessageContainer v_NAS_TxMessageContainer_SMS;

    var CP_PDU_Type v_RcvSMS_CP_PDU;

    var template CP_PDU_Type v_ExpSMS_CP_PDU;

    var bitstring v_NAS_RcvBitsMsgContainer_SMS;

    var B3_Type v_TI_Value;

    var TransactionIdentifier_Type v_MO_TI_FromSS;

    var TransactionIdentifier_Type v_MO_TI_FromUE;

    var O1_Type v_RpMsgRefRcvd;

    var SMS_DELIVER_Type v_SMS_DELIVER_Sent;

    var SMS_SUBMIT_Type v_SMS_SUBMIT_Rcvd;
    var octetstring v_PRBS_SMS := f_GetN_OctetsFromPRBS (0, 123); // 123 fits with the requirement from 36.509

    var integer v_LoopDelay := 60;

    timer t_LoopDelay;
...

    //@siclog "Step 16a6" siclog@

    t_LoopDelay.timeout;

    //@siclog "Step 16a7" siclog@

    // 1st repetition

        v_Rxd_SMS := f_NBIOT_508Check_CP_MOSMSDataTransfer (nbiot_Cell50);  

    decvalue(oct2bit(v_Rxd_SMS), v_RcvSMS_CP_PDU);  //This decodes the received payload to CP PDU level

    v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT;

    if (not (match (v_SMS_DELIVER_Sent.tP_UserData, v_SMS_SUBMIT_Rcvd.tP_UserData))) {

      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed TPDU does not match");

    }

    else {

      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a7");

    }
    //@siclog "Step 16a7A" siclog@  
    // The SS transmits a CP-ACK encapsulated in a Downlink NAS Transport message.

    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;

    v_MO_TI_FromSS.tI_Flag := '1'B;

    v_MO_TI_FromSS.tI_Value := v_TI_Value;

    v_MO_TI_FromUE.tI_Flag := '0'B;

    v_MO_TI_FromUE.tI_Value := v_TI_Value;

    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;

    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (v_MO_TI_FromSS)));

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          v_NAS_TxMessageContainer_SMS))));

    //@siclog "Step 16a7B" siclog@ 

    // The SS transmits a CP-DATA containing an RP-ACK RPDU encapsulated in an Downlink NAS transport message

    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (v_MO_TI_FromSS,

                                                                                                cs_CP_UserData_RP_ACK_dl(v_RpMsgRefRcvd))));

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          v_NAS_TxMessageContainer_SMS))));

    //@siclog "Step 16a7C" siclog@ 

    // Check: Does the UE transmit a CP-ACK encapsulated in an Uplink NAS Transport message?

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                   tsc_SRB1bis,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;

    v_NAS_RcvMsgContainer_SMS :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);

    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (v_MO_TI_FromUE);

    v_RcvSMS_CP_PDU := f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test case 22.1.1", "Step 9C");

    //@siclog "Step 16a8" siclog@

    // 2nd repetition

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                           tsc_SRB1bis,

                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                             cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                               cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;
    // Note TI value used by the UE and set TI variables accordingly

    v_NAS_RcvMsgContainer_SMS :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);

    decvalue(v_NAS_RcvBitsMsgContainer_SMS, v_RcvSMS_CP_PDU);

    v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT;

    if (not (match (v_SMS_DELIVER_Sent.tP_UserData, v_SMS_SUBMIT_Rcvd.tP_UserData))) {

      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed TPDU does not match");

    }

    else {

      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a8");

    }
    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;

    v_MO_TI_FromSS.tI_Flag := '1'B;

    v_MO_TI_FromSS.tI_Value := v_TI_Value;

    v_MO_TI_FromUE.tI_Flag := '0'B;

    v_MO_TI_FromUE.tI_Value := v_TI_Value;

    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;

    //@siclog "Step 16a8A" siclog@
MCC160 Implementation

After  change

    function f_TC_22_1_1_Module3() runs on NBIOT_PTC

  {

    var  octetstring v_SMS_Message;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var octetstring v_Rxd_SMS;

    var NAS_MessageContainer v_NAS_RcvMsgContainer_SMS;

    var template (value) NAS_MessageContainer v_NAS_TxMessageContainer_SMS;

    var CP_PDU_Type v_RcvSMS_CP_PDU;

    var template CP_PDU_Type v_ExpSMS_CP_PDU;

    var bitstring v_NAS_RcvBitsMsgContainer_SMS;

    var B3_Type v_TI_Value;

    var TransactionIdentifier_Type v_MO_TI_FromSS;

    var TransactionIdentifier_Type v_MO_TI_FromUE;

    var O1_Type v_RpMsgRefRcvd;

    var SMS_DELIVER_Type v_SMS_DELIVER_Sent;

    var SMS_SUBMIT_Type v_SMS_SUBMIT_Rcvd;

    var octetstring v_PRBS_SMS := f_GetN_OctetsFromPRBS (0, 124); // 124 fits with the requirement from 36.509 @sic R5s180287 sic@

    var integer v_LoopDelay := 60;

    timer t_LoopDelay;

    //@siclog "Step 16a1" siclog@

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell50);

    //@siclog "Step 16a1A- 16a1B" siclog@

    f_NBIOT_Activate_TestMode (nbiot_Cell50, tsc_UE_TestLoopMode_TypeH);

    //@siclog "Step 16a2 - 16a3" siclog@

    f_NBIOT_CloseUE_TestLoopModeGH (nbiot_Cell50, tsc_UE_TestLoopMode_TypeH, tsc_TestLoopModeGH_UplinkMode_NAS, 2, v_LoopDelay);

    t_LoopDelay.start(f_NBIOT_SetTimerToleranceMin(nasTimer, int2float(v_LoopDelay)));

    //@siclog "Step 16a4" siclog@

    //The SS transmits a CP-ACK encapsulated in a Downlink NAS transport message to the UE.

    v_SMS_DELIVER_Sent := valueof(cs_SMS_DELIVER_IoT (v_PRBS_SMS)); // data TPDU with random data

    v_SMS_Message := f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (cs_MT_TI0_FromSS,

                                                        cs_CP_UserData_DELIVER_IoT(v_SMS_DELIVER_Sent))); // data TPDU with random data

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          cs_NAS_MsgContainer_SMS (v_SMS_Message)))));

    //@siclog "Step 16a4A" siclog@

    // The UE transmits a CP-ACK encapsulated in an Uplink NAS transport message.

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                    tsc_SRB1bis,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                               cr_NAS_MsgContainer_SMS (?) )))) -> value v_ReceivedAsp;

    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);

    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (cr_MT_TI0_FromUE);

    f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test Case 22.1.1 M3", "Step 16a4A"); // @sic R5s180287 sic@

    //@siclog "Step 16a4B" siclog@

    //Check: Does the UE transmit a CP-DATA containing an RP-ACK RPDU encapsulated in an Uplink NAS transport message?

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                   tsc_SRB1bis,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;

    // Store procedure transaction id to be used in SS responses.

    // needs to be decoded with Decvalue !!

    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    v_ExpSMS_CP_PDU := cr_CP_DATA_PDU (cr_MT_TI0_FromUE, cr_CP_UserData_RP_ACK_ul);    

    f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test Case 22.1.1 M3", "Step 16a4B");

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a4B");
    //@siclog "Step 16a4C" siclog@

    //The SS transmits a CP-ACK encapsulated in a Downlink NAS transport message to the UE.

    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (cs_MT_TI0_FromSS)));

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          v_NAS_TxMessageContainer_SMS))));

    //@siclog "Step 16a5" siclog@

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell50);

    //@siclog "Step 16a6" siclog@

    t_LoopDelay.timeout;

    //@siclog "Step 16a7" siclog@

    // 1st repetition

    v_Rxd_SMS := f_NBIOT_508Check_CP_MOSMSDataTransfer (nbiot_Cell50, v_SMS_DELIVER_Sent.tP_UserData); // @sic R5s180287 sic@   

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_Rxd_SMS);

    if (decvalue(v_NAS_RcvBitsMsgContainer_SMS, v_RcvSMS_CP_PDU) != 0) { // @sic R5s180287 sic@

      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed SMS does not match");

    }

    v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT; // @sic R5s180287 sic@

    if (not (match (v_SMS_DELIVER_Sent.tP_UserData, v_SMS_SUBMIT_Rcvd.tP_UserData))) {

      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed TPDU does not match");

    }

    else {

      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a7");

    }
    //@siclog "Step 16a7A" siclog@

    // The SS transmits a CP-ACK encapsulated in a Downlink NAS Transport message.

    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;

    v_MO_TI_FromSS.tI_Flag := '1'B;

    v_MO_TI_FromSS.tI_Value := v_TI_Value;

    v_MO_TI_FromUE.tI_Flag := '0'B;

    v_MO_TI_FromUE.tI_Value := v_TI_Value;

    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;

    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (v_MO_TI_FromSS)));

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          v_NAS_TxMessageContainer_SMS))));

    //@siclog "Step 16a7B" siclog@

    // The SS transmits a CP-DATA containing an RP-ACK RPDU encapsulated in an Downlink NAS transport message

    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (v_MO_TI_FromSS,

                                                                                                cs_CP_UserData_RP_ACK_dl(v_RpMsgRefRcvd))));

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          v_NAS_TxMessageContainer_SMS))));

    //@siclog "Step 16a7C" siclog@

    // Check: Does the UE transmit a CP-ACK encapsulated in an Uplink NAS Transport message?

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                   tsc_SRB1bis,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;

    v_NAS_RcvMsgContainer_SMS :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);

    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (v_MO_TI_FromUE);

    v_RcvSMS_CP_PDU := f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test case 22.1.1", "Step 9C");

    //@siclog "Step 16a8" siclog@

    // 2nd repetition

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,

                                           tsc_SRB1bis,

                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                             cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                               cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;  // @sic R5s180287 sic@
    // Store procedure transaction id to be used in SS responses.

    // needs to be decoded with Decvalue !!

    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);

    if (decvalue(v_NAS_RcvBitsMsgContainer_SMS, v_RcvSMS_CP_PDU) != 0) { // @sic R5s180287 sic@

      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed SMS does not match");

    }

    v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT; // @sic R5s180287 sic@

    if (not (match (v_SMS_DELIVER_Sent.tP_UserData, v_SMS_SUBMIT_Rcvd.tP_UserData))) {

      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed TPDU does not match");

    }

    else {

      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a8");

    }
    // Note TI value used by the UE and set TI variables accordingly

    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;

    v_MO_TI_FromSS.tI_Flag := '1'B;

    v_MO_TI_FromSS.tI_Value := v_TI_Value;

    v_MO_TI_FromUE.tI_Flag := '0'B;

    v_MO_TI_FromUE.tI_Value := v_TI_Value;

    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;

    //@siclog "Step 16a8A" siclog@

    // The SS transmits a CP-ACK encapsulated in a Downlink NAS Transport message.

    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (v_MO_TI_FromSS)));

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,

                                   tsc_SRB1bis,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,

                                                                          v_NAS_TxMessageContainer_SMS))));

    //@siclog "Step 16a8B" siclog@

…

Change 4

	Function name
	f_NBIOT_508Check_CP_MOSMSDataTransfer

	Reason for change
	1. As given the reason in change 3.1 the received SMS cannot be matched on the CP PDU but only on the TP PDU. For this reason and also because it will simplify the code we propose to move the check required in step 16a8 outside this function. This means the smsmessage in the NAS message container will be allowed ANY value
2. The above change implies that the second input parameter of the modified function is not needed anymore.
3. The IEI for smsmessage in the NAS message container in the ControlPlaneServiceRequest is given by 24.301 as 67’O – for this reason another template with correct IEI needs to be created and used.

	Summary of change
	1. The smsmessage in NAS message container in the CPSR is allowed for ANY value.
2. Removed second input parameter from function definition.
3. Newly created template for smsmessage is used when it is received as part of CPSR.

	TTCN module
	NBIOT_CommonProcedures.ttcn

	MCC160 Comment
	1. Check modified. See MCC160 implementation below.
2. Not done as the generic procedure needs to optionally check the TPDU. Parameter changed to TPDU.

3. Not needed. Existing template was parametrized. See below.


Before  change

  function f_NBIOT_508Check_CP_MOSMSDataTransfer(NBIOT_CellId_Type p_CellId,

                                                 template (present) octetstring p_DataExpected) runs on NBIOT_PTC return octetstring

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var AccessPointName v_APN;

    var template (value) PDN_Address v_PDN_Address;

    var integer v_PDNIndex := 0;

    var boolean v_ConfigurePDN := false;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    timer t_WaitForPDN := 2.0;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;

    }

    if (px_DoAttachWithoutPDN) {

      v_AttachWithoutPDN := true_;

    }

    v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                      v_PDNType,

                                                                      -,

                                                                      v_HeaderCompressConfigRX);

    v_NasInd_Expected := cr_NAS_Indication(tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_CPServiceType(-, '000'B), cr_NAS_MsgContainer_SMS(p_DataExpected))); // @sic R5-174456 sic@

    // Step 1

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, mo_Data);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO SMS Data Transfer Step1");

    // Step 2

    f_NBIOT_RRC_ConnectionSetup_Def();

    // Step 3

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO SMS Data Transfer Step3b1");

    // Step 3b2

    t_WaitForPDN.start;

    alt {

      [px_nonSMSTransport_CP_CIoT and px_DoAttachWithoutPDN] SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                                                                                 tsc_SRB1bis,

                                                                                                 cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                                                                                   v_PDNConnectivityReq_Expected))) -> value v_ReceivedAsp

        {  // Step 3b3a1

          v_PDNConnectivityReq_Rxd := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;

          v_ConfigurePDN := true;

          // Step 3b3a2

          t_WaitForPDN.stop;

        }

      // Step 3b3b1

      [] t_WaitForPDN.timeout {}

    }

    if (v_ConfigurePDN) {

      v_PDNIndex := f_NBIOT_MobileInfo_GetNoOfPDNsConfigured();

      f_NBIOT_MobileInfo_SetAssignedPdnAndPTI (v_PDNIndex,

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue,

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier);

      v_PcoFromUE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions;

      v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE));

      f_NBIOT_MobileInfo_SetExtPCO (v_PDNIndex, v_PcoFromUE);

      if (ispresent (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName)) {

        v_APN := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName;

        v_APN.iei := omit;

      } else {

        v_APN := valueof(cs_AccessPointName(px_AccessPointName));

      }

      f_NBIOT_MobileInfo_SetAPN (v_PDNIndex, v_APN); // NOTE: APN is not omit anymore

      v_PDN_Address := f_CIOT_GetPDNAddress(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE)); // @sic R5s170597 sic@

      // Step 3b3a3

      SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                     tsc_SRB1bis,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                              f_NBIOT_MobileInfo_GetPTI(v_PDNIndex),

                                                                                              v_APN,

                                                                                              v_PDN_Address,

                                                                                              cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),

                                                                                              -,

                                                                                              -,

                                                                                              -,

                                                                                              v_HeaderCompressConfigTX,
                                                                                              cs_ControlPlaneInd,

                                                                                              cs_ServingPLMNRateControl(int2oct (4, 2))))));      // Step 4a1b3a4

      // Step 3b3a4

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId))));

      if (pc_IP_PDN) {

        f_NBIOT_DelayForUserPlaneSignalling_CP(p_CellId);

      }

    }

    // Step 4

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_SERVICE_ACCEPT)));

    return v_NasInd_Rxd.Pdu.Msg.cONTROL_PLANE_SERVICE_REQUEST.nasMessage.smsmessage;

    }
After  change

  function f_NBIOT_508Check_CP_MOSMSDataTransfer(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return octetstring

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var AccessPointName v_APN;

    var template (value) PDN_Address v_PDN_Address;

    var integer v_PDNIndex := 0;

    var boolean v_ConfigurePDN := false;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;
    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    timer t_WaitForPDN := 2.0;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;
      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;
    }

    if (px_DoAttachWithoutPDN) {

      v_AttachWithoutPDN := true_;

    }

    v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                      v_PDNType,

                                                                      -,

    v_NasInd_Expected := cr_NAS_Indication(tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_CPServiceType(-, '000'B), cr_NAS_MsgContainer_SMS_CPSR(?))); // @sic R5-174456 sic@

    // Step 1

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, mo_Data);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO SMS Data Transfer Step1");

    // Step 2

    f_NBIOT_RRC_ConnectionSetup_Def();

    // Step 3

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO SMS Data Transfer Step3b1");

    // Step 3b2

    t_WaitForPDN.start;

    alt {

      [px_nonSMSTransport_CP_CIoT and px_DoAttachWithoutPDN] SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                                                                                 tsc_SRB1bis,

                                                                                                 cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                                                                                   v_PDNConnectivityReq_Expected))) -> value v_ReceivedAsp

        {  // Step 3b3a1

          v_PDNConnectivityReq_Rxd := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;

          v_ConfigurePDN := true;

          // Step 3b3a2

          t_WaitForPDN.stop;

        }

      // Step 3b3b1

      [] t_WaitForPDN.timeout {}

    }

    if (v_ConfigurePDN) {

      v_PDNIndex := f_NBIOT_MobileInfo_GetNoOfPDNsConfigured();

      f_NBIOT_MobileInfo_SetAssignedPdnAndPTI (v_PDNIndex,

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue,

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier);

      v_PcoFromUE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions;

      v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE));

      f_NBIOT_MobileInfo_SetExtPCO (v_PDNIndex, v_PcoFromUE);

      if (ispresent (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName)) {

        v_APN := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName;

        v_APN.iei := omit;

      } else {

        v_APN := valueof(cs_AccessPointName(px_AccessPointName));

      }

      f_NBIOT_MobileInfo_SetAPN (v_PDNIndex, v_APN); // NOTE: APN is not omit anymore

      v_PDN_Address := f_CIOT_GetPDNAddress(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE)); // @sic R5s170597 sic@

      // Step 3b3a3

      SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                     tsc_SRB1bis,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                              f_NBIOT_MobileInfo_GetPTI(v_PDNIndex),

                                                                                              v_APN,

                                                                                              v_PDN_Address,

                                                                                              cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),

                                                                                              -,

                                                                                              -,

                                                                                              -,

                                                                                              v_HeaderCompressConfigTX,
                                                                                              cs_ControlPlaneInd,

                                                                                              cs_ServingPLMNRateControl(int2oct (4, 2))))));      // Step 4a1b3a4

      // Step 3b3a4

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId))));

      if (pc_IP_PDN) {

        f_NBIOT_DelayForUserPlaneSignalling_CP(p_CellId);

      }

    }

    // Step 4

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_SERVICE_ACCEPT)));

    return v_NasInd_Rxd.Pdu.Msg.cONTROL_PLANE_SERVICE_REQUEST.nasMessage.smsmessage;

    }
MCC160 Implementation

	Function name
	f_NBIOT_508Check_CP_MOSMSDataTransfer

	Reason for change
	TPDU_Expected is to be checked if provided in 36.508 cl. 8.1.5A.3A

	Summary of change
	Parameter was changed to TPDU and check was performed optionally.


After  change

  function f_NBIOT_508Check_CP_MOSMSDataTransfer(NBIOT_CellId_Type p_CellId,

                                                 template (omit) octetstring p_TPDU_Expected) runs on NBIOT_PTC return octetstring  // @sic R5s180287 sic@

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var octetstring v_Rxd_SMS;

    var NAS_MessageContainer v_NAS_RcvMsgContainer_SMS;

    var template (value) NAS_MessageContainer v_NAS_TxMessageContainer_SMS;

    var CP_PDU_Type v_RcvSMS_CP_PDU;

    var template CP_PDU_Type v_ExpSMS_CP_PDU;

    var bitstring v_NAS_RcvBitsMsgContainer_SMS;

    var SMS_DELIVER_Type v_SMS_DELIVER_Sent;

    var SMS_SUBMIT_Type v_SMS_SUBMIT_Rcvd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var AccessPointName v_APN;

    var template (value) PDN_Address v_PDN_Address;

    var integer v_PDNIndex := 0;

    var boolean v_ConfigurePDN := false;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    timer t_WaitForPDN := 2.0;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;

    }

    if (px_DoAttachWithoutPDN) {

      v_AttachWithoutPDN := true_;

    }

    v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                      v_PDNType,

                                                                      -,

                                                                      v_HeaderCompressConfigRX);

    v_NasInd_Expected := cr_NAS_Indication(tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_CPServiceType(-, '000'B), cr_NAS_MsgContainer_SMS(?, '67'O))); // @sic R5-174456 sic@  @sic R5s180287 sic@
    // Step 1

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, mo_Data);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO SMS Data Transfer Step1");

    // Step 2

    f_NBIOT_RRC_ConnectionSetup_Def();

    // Step 3

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    // Step 3b1

    v_NAS_RcvMsgContainer_SMS := v_NasInd_Rxd.Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);

    if (decvalue(v_NAS_RcvBitsMsgContainer_SMS, v_RcvSMS_CP_PDU) != 0) {  // @sic R5s180287 sic@

      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed SMS does not match");

    }

    if (isvalue(p_TPDU_Expected)) {  // @sic R5s180287 sic@

      // TPDU is provided as parameter => is to be checked against TPDU received in SMS SUBMIT !!

      v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT;

      if (not (match (valueof(p_TPDU_Expected), v_SMS_SUBMIT_Rcvd.tP_UserData))) {

        f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "508Check TPDU Data Transfer Step3b1");

      }  

      else {

        f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check TPDU Data Transfer Step3b1");

      }  

    } 

    else {

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO SMS Data Transfer Step3b1");

    }     

    // Step 3b2

    t_WaitForPDN.start;
…

Change 5

	Template name
	cr_NAS_MsgContainer_SMS_CPSR

	Reason for change
	Based on the Change 4.3 new template needs to be created which is then used in the f_NBIOT_508Check_CP_MOSMSDataTransfer.

	Summary of change
	New template created.

	TTCN module
	EPS_NAS_Templates.ttcn

	MCC160 Comment
	Not needed. Paramterized cr_NAS_MsgContainer_SMS as shown below.
Usage shown in Change 3 above.


After  change

template (present) NAS_MessageContainer cr_NAS_MsgContainer_SMS_CPSR(template (present) octetstring p_SMS_Message) :=

  { 

    iei         := '67'O,

    iel         := ?,

    smsmessage  := p_SMS_Message

  };  

MCC160 Implementation

  template (present) NAS_MessageContainer cr_NAS_MsgContainer_SMS(template (present) octetstring p_SMS_Message, 

                                                                  template (omit) IEI8_Type p_IEI := omit) :=

  { /* @status    APPROVED (LTE) */

    iei         := p_IEI,
    iel         := ?,

    smsmessage  := p_SMS_Message

  };
Change 6

	Function name
	f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_4

	Reason for change
	In case the UE requested Attach without PDN (px_DoAttachWithoutPDN = TRUE) the EPSbearerContextStatus as signalled in the CPSR will be omitted. To account for that there needs to different check for context status based on the pix value.

	Summary of change
	Added variable v_EBI5_7 which is populated differently based on the px_DoAttachWithoutPDN.

	TTCN module
	NBIOT_CommonProcedures.ttcn

	MCC160 Comment
	Rejected. The current template already contains ‘ifpresent’.


Before  change

  function f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_4(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return template (omit) NAS_UL_Message_Type

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var boolean v_ConfigurePDN := false;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    timer t_WaitForPDN := 2.0;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

    }

    v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                      v_PDNType,

                                                                      cr_AccessPointName_Any,

                                                                      v_HeaderCompressConfigRX);

    v_NasInd_Expected := (cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?), cr_ESM_DATA_TRANSPORT (?, ?)),  // @sic R5-174456 sic@

                          cr_NAS_Indication (tsc_SHT_IntegrityProtected, cr_CONTROL_PLANE_SERVICE_REQUEST(?)));
    // Step 2

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, mt_Access);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP ResponseToPagingForMTAccess Step2");

    // Step 3
After  change

  function f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_4(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return template (omit) NAS_UL_Message_Type

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var boolean v_ConfigurePDN := false;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    var B3_Type v_EBI5_7 := '001'B;
    timer t_WaitForPDN := 2.0;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

    }

    v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                      v_PDNType,

                                                                      cr_AccessPointName_Any,

                                                                      v_HeaderCompressConfigRX);

    if (px_DoAttachWithoutPDN) {

      v_EBI5_7 := '000'B;

    }
    v_NasInd_Expected := (cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?,-,-,cr_508_EPS_BearerContextStatusTAUReq(v_EBI5_7) ifpresent), cr_ESM_DATA_TRANSPORT (?, ?)),  // @sic R5-174456 sic@

                          cr_NAS_Indication (tsc_SHT_IntegrityProtected, cr_CONTROL_PLANE_SERVICE_REQUEST(?,-,-,cr_508_EPS_BearerContextStatusTAUReq(v_EBI5_7) ifpresent)));
    // Step 2

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, mt_Access);

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP ResponseToPagingForMTAccess Step2");

    // Step 3
Additional changes from MCC160

	Template name
	cs_SMS_DELIVER_IoT

	Reason for change
	There is an error in length calculation

	Summary of change
	Error was corrected.

	TTCN module
	CIoT_SMS_Templates.ttcn


After  change

  template (value) SMS_DELIVER_Type cs_SMS_DELIVER_IoT (octetstring p_PRBS) :=

  { /* @status     */

    // according to 36.509 test loop mode H description the TPDU shall consist of 142 octets

    // with a TP address of 10 digits we have 1 + 7 + 1 + 1 + 7 + 1 = 18 octets plus PRBS => PRBS shall be 124 octets

    // calculated in 7 bit chars this means 140 7bit chars + padding

    tP_ReplyPath := '0'B,

    tP_UserDataHeaderIndicator := '0'B,

    tP_StatusReportIndication := '0'B,

    spare2 := '00'B,

    tP_MoreMessagesToSend := '1'B,

    tP_MessageTypeIndicator := tsc_MT_SMS_DELIVER,

    tP_OriginatingAddress := cs_TP_Address ('2233445566'O),

    tP_ProtocolIdentifier := cs_TP_ProtocolIdentifier,

    tP_DataCodingScheme_Type := cs_TP_DataCodingScheme,

    tP_ServiceCentreTimeStamp := f_BCD_TimestampWithTimezone(),

    tP_UserDataLength := int2oct(140,1),

    tP_UserData := p_PRBS // 124 octets

  };
Additional change from MCC160

	Function name
	f_TC_23_1_2_EUTRA

	Reason for change
	There is an error in length calculation of v_PRBS_SMS which is set to 123

	Summary of change
	The error in length calculation of v_PRBS_SMS which was corrected. Set to 124.

	TTCN module
	CIoT_SMS_Templates.ttcn


After  change

    function f_TC_23_1_2_EUTRA() runs on EUTRA_PTC

  {

    var EUTRA_SecurityParams_Type v_AuthParams;

    var SRB_COMMON_IND v_ReceivedAsp;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_MessageContainer v_NAS_RcvMsgContainer_SMS;

    var template (value) NAS_MessageContainer v_NAS_TxMessageContainer_SMS;

    var CP_PDU_Type v_RcvSMS_CP_PDU;

    var template CP_PDU_Type v_ExpSMS_CP_PDU;

    var bitstring v_NAS_RcvBitsMsgContainer_SMS;

    var B3_Type v_TI_Value;

    var TransactionIdentifier_Type v_MO_TI_FromSS;

    var TransactionIdentifier_Type v_MO_TI_FromUE;

    var O1_Type v_RpMsgRefRcvd;

    var SMS_DELIVER_Type v_SMS_DELIVER_Sent;

    var SMS_DELIVER_Type v_SMS_DELIVER_Rcvd;

    var template CP_UserData_Type v_Exp_CP_UserData;

    var octetstring v_PRBS_SMS;

    timer t_LoopDelay;

    if ((px_SMSTransport_CP_CIoT != true)){

      FatalError(__FILE__, __LINE__, "UE not configured correctly for this test case");

    }

    v_PRBS_SMS := f_GetN_OctetsFromPRBS (0, 124); // 124 fits with the requirement from 36.509
….
5. Execution Log Files

5.1 Qualcomm MDM9206
The Qualcomm UE passed this NB-IoT in LTE FDD band 1 on Anritsu Protocol Conformance Test System ME7834LA.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

